Contents: Basic course G3 and Upgrade course TNC 426/430

Contents: Basic course G3 and Upgrade course TNC 426/430

Task: **Holes**

Program(s): _____



Program layout: **Holes**

Begin program	PGM MGT	*BEGINN PGM . . . MM*
Define workpiece blank		*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Define tool	TOOL DEF	*TOOL DEF . . . L . . . R . . .*
Call tool data	TOOL CALL	*TOOL CALL . . . . . S . . .*
Move to clearance height	L	*L Z+100 R0 FMAX*
Move to starting point	L	*L X . . . Y . . . R0 FMAX M3*
Move to set up clearance	L	*L Z+ . . . R FMAX M8*
Drill	L	*L Z– . . . R F100 M*
Retract drill	L	*L Z+2 R0 FMAX*
Move to next position	L	*L X . . . Y . . . FMAX*
Drill	L	*L Z– . . .*
Retract drill	L	*L Z+ . . . FMAX*
Next hole	L	*L X . . . Y . . . R0 FMAX*
Drill	L	*L Z–22 R0*
Return to clearance height End of program	L	*L Z+100 R0 FMAX M2 (M30)*

Solution: **Holes**



Complete program

```
0  BEGIN PGM 151 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3  TOOL DEF 1 L+0 R+4 ..... DEFINE TOOL
4  TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5  L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6  L X+20 Y+30 R0 F MAX M3 ..... FIRST HOLE
7  L Z+2 R0 F MAX M8
8  L Z-22 R0 F400 ..... DRILL
9  L Z+2 R0 F MAX
10 L X+50 Y+70 R0 F MAX ..... SECOND HOLE
11 L Z-22 R0 F400
12 L Z+2 R0 F MAX
13 L X+75 Y+30 R0 F MAX ..... THIRD HOLE
14 L Z-22 R0 F400
15 L Z+100 R0 F MAX M2
16 END PGM 151 MM
```

Task: **Square**

Program(s): _____



Begin program		*BEGIN PGM . . . MM*
Define workpiece blank		*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .*
		BLK-FORM 0.2 X . . . Y . . . Z . . .
Define tool		*TOOL DEF . . . L . . . R . . .*
Call tool data		*TOOL CALL . . . . . S . . .*
Move to clearance height		*L Z+ . . . R0 F . . .*
Move to auxiliary point	R0	*L X . . . Y . . . R . . . F . . .*
Plunging depth		*L Z . . . R . . . F . . . M . . .*
Contour starting point	RL/RR	*L X . . . Y . . . RR/RL*
Contour coordinates	RL/RR	*L X . . . Y . . . RR/RL*
⋮		⋮
Last contour point	RL/RR	*L X . . . Y . . . RR/RL*
Move to auxiliary point	R0	*L X . . . Y . . . R . . .*
Retract tool, PGM end		*L Z . . . R . . . F . . . M . . .*

Solution: **Square**



Complete program

```
0  BEGIN PGM 152 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3  TOOL DEF 1 L+0 R+8 ..... DEFINE TOOL
4  TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5  L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6  L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT (R0)
7  L Z-5 R0 F MAX M3 ..... PLUNGING DEPTH
8  L X+0 Y+50 RL F400 ..... CONTOUR START. POINT (RL/RR)
9  L X+50 Y+100
10 L X+100 Y+50
11 L X+50 Y+0
12 L X+0 Y+50 RL ..... LAST CONTOUR POINT
13 L X-30 R0 F MAX M5 ..... AUXILIARY POINT
14 L Z+100 R0 F MAX M2 ..... RETRACT TOOL/PGM END
15 END PGM 152 MM
```

Task: **Rounding / chamfering corners** Program(s): _____



Solution: **Rounding / chamfering corners**



Complete program

```
0 BEGIN PGM 153 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3 TOOL DEF 1 L+0 R+8 ..... DEFINE TOOL
4 TOOL CALL 1 Z S4000 ..... CALL TOOL DATA
5 L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6 L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT (R0)
7 L Z-5 R0 F MAX M3
8 L X+0 Y+50 RL F200 ..... CONTOUT STARTING POINT RL
9 L X+50 Y+100
10 RND R10 ..... ROUND CORNERS
11 L X+100 Y+50
12 L X+50 Y+0
13 CHF 5 ..... CHAMFER
14 L X+0 Y+50 RL
15 L X-30 R0 M5 ..... AUXILIARY POINT (RO)
16 L Z+100 R0 F MAX M2 ..... PGM END
17 END PGM 153 MM
```

Task: **Rounding corners** Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Define tool	*TOOL DEF . . . L . . . R . . .*
Call tool data	*TOOL CALL . . . . . S . . .*
Move to clearance height	*L Z . . .*
Auxiliary point anfahren	*L X . . . Y . . .*
Plunging depth	*L Z . . .*
Approach tangentially	*APPR . . .*
Contour	*L . . .*
Depart tangentially	*DEP . . .*
Retract tool, PGM end	*L Z . . .*

Solution: **Rounding corners**



Complete program

```
0 BEGIN PGM 154 MM
1 BLK FORM 0.1 Z X-20 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL DEF 1 L+0 R+8
4 TOOL CALL 1 Z S4000
5 L Z+100 R0 F MAX
6 L X-30 Y+70 R0 F MAX ..... AUXILIARY POINT (R0)
7 L Z-5 R0 F MAX M3
8 APPR LCT X+10 Y+70 R5 RL F400 ..... APPROACH STARTING POINT
                                      TANGENTIALLY
9 L X+10 Y+90
10 RND R10
11 L X+50 Y+90
12 L Y+50 X+90
13 L X+90 Y+10
14 RND R10
15 L X+50 Y+10
16 L X+10 Y+50
17 L Y+70 ..... LAST CONTOUR POINT RL
18 DEP LCT X-30 Y+70 R5 ..... DEPART TANGENTIALLY TO
                              AUXILIARY POINT
19 L Z+100 R0 F MAX M2
20 END PGM 154 MM
```

Task: **Linear movements**

Program(s): _____



Solution: **Linear movements**



Complete program

```
0  BEGIN PGM 250 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0 ..... WORKPIECE BLANK DEFINITION
3  TOOL CALL 7 Z S2500 ..... CALL TOOL DATA; R4
4  L Z+100 R0 F9999 ..... CLEARANCE HEIGHT
5  L X-20 Y+40 ..... AUXILIARY POINT (R0)
6  L Z+2 M3
7  L Z-5 F1000
8  APPR LCT X+5 Y+40 R3 RL F300 ..... CONTOUR STARTING POINT
9  L Y+90 (APPROACH TANGENTIALLY)
10 CHF 20
11 L X+60
12 L X+90 Y+80
13 L Y+40
14 CHF 10
15 L X+60 Y+10
16 L X+5
17 RND R10
18 L Y+40 ..... LAST CONTOUR POINT
19 DEP LCT X-20 Y+40 R3 ..... AUXILIARY POINT (R0)
20 L Z+100 R0 F MAX M2 ..... RETRACT TOOL
21 END PGM 250 MM
```

Task: **Circular movements**

Program(s): _____





Complete program

```
0  BEGIN PGM 251 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 7 Z S2500 ..... R4
4  L Z+100 R0 F9999
5  L X+20 Y-20 ..... AUXILIARY POINT (R0)
6  L Z+2 M3
7  L Z-5 F500
8  APPR LCT X+20 Y+30 R3 RL F300 ..... CONTOUR STARTING POINT
9  L X+0 (APPROACH TANGENTIALLY)
10 RND R4
11 L X+15 Y+45
12 CR X+15 Y+60 R+20 DR+
13 L X+0 Y+75
14 CR X+20 Y+95 R+20 DR-
15 L X+40
16 CT X+65 Y+80
17 CC X+75 Y+80
18 C X+85 Y+80 DR+
19 L X+95
20 RND R5
21 L Y+50
22 L X+75 Y+30
23 RND R8
24 L Y+20
25 CC X+60 Y+20
26 C X+45 Y+20 DR-
27 L Y+30
28 RND R9
29 L X+20 ..... LAST CONTOUR POINT
30 DEP LCT X+20 Y-20 R3 F500 ..... AUXILIARY POINT (R0)
31 L Z+100 R0 F MAX M2
32 END PGM 251 MM
```

Task: **Circular arc with CC, C**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Call tool data	*TOOL CALL . . . . . S . . .*
Move to clearance height	*L . . .*
Move to auxiliary point R0	*L . . . R0*
Plunging depth	*L . . .*
Approach contour tangentially	*APPR . . . RL/RR*
Define circle center	*CC . . .*
Circular movement	*C . . .*
Depart tangentially (aux.pt.)	*DEP . . .*
Retract tool, PGM end	L . . .

Solution: **Circular arc with CC, C**



Complete program

```
0 BEGIN PGM 206 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y+50 R0 F MAX ..... AUXILIARY POINT
6 L Z-5 R0 F MAX M3
7 APPR LT X+0 Y+20 LEN10 RL F250 M8
8 L X+23,542 RL
9 CC X+50 Y+50 ..... CIRCLE CENTER
10 C Y+80 X+23,542 DR+ ..... CIRCULAR MOVEMENT
11 L X+0 RL
12 DEP LT LEN10 ..... DEPART TANGENTIALLY (AUX.PT.)
13 L Z+100 R0 F MAX M2
14 END PGM 206 MM
```

Task: **Tangential contour connection (cartesian)**

Program(s): _____



Solution: **Tangential contour connection (cartesian)**



Complete program

```
0  BEGIN PGM 207 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 11 Z S2500 ..... R10
4  L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
5  L X+45 Y-25 R0 F MAX ..... AUXILIARY POINT
6  L Z-5 F MAX M13
7  APPR LT X+20 Y+0 LEN5 RR F250 ..... APPROACH STARTING POINT OF
8  L Y+30 CONTOUR TANGENTIALLY
9  CT X+70 Y+60 ..... TANGENTIAL CIRCULAR PATH
10 L X+100
11 DEP LT LEN5
12 L Z+100 R0 F MAX M2
13 END PGM 207 MM
```

Task: **Circular arcs**

Program(s): _____



Point	X	Y
P_1	6,645	35,495
P_2	55,505	69,488

Point	X	Y
P_3	58,995	30,025
P_4	19,732	21,191

Solution: **Circular arcs**



Complete program

```
0  BEGIN PGM 208 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 13 Z S2500 ..... R20
4  L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
5  L X-30 Y+30 R0 F MAX M3 ..... AUXILIARY POINT
6  L Z-5 F MAX ..... DEPTH
7  APPR LCT X+5 Y+30 R5 RL F250 M8 ..... APPROACH TANGENTIALLY
8  CC X+15 Y+30 ..... CIRCLE CENTER
9  C X+6,645 Y+35,495 DR- ..... CIRCLE
10 CT X+55,505 Y+69,488 ..... TANGENTIAL CIRCULAR PATH
11 CC X+60 Y+50
12 C X+58,995 Y+30,025 DR-
13 CT X+19,732 Y+21,191
14 CC X+15 Y+30
15 C X+5 Y+30 DR-
16 DEP LCT X-30 Y+30 R5 ..... DEPART TANGENTIALLY
17 L Z+100 R0 F MAX M2
18 END PGM 208 MM
```

Task: **Circular arc with CR**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .*
	BLK-FORM 0.2 X . . . Y . . . Z . . .
Call tool data	*TOOL CALL . . . . . S . . .*
Move to clearance height	*L . . .*
Move to auxiliary point	*. . .*
Plunging depth	*. . .*
Approach tangentially	*. . .*
Contour	*. . .*
Depart tangentially	*. . .*
Retract tool, PGM end	*. . .*

Solution: **Circular arc with CR**



Complete program

```
0 BEGIN PGM 209 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M13 ..... AUXILIARY POINT
6 L Z-5 F MAX
7 APPR LT X+10 Y+0 LEN10 RL F250 ..... APPROACH STARTING POINT
                                       TANGENTIALLY
8 L Y+50
9 CR X+90 Y+50 R+80 DR- ..... CIRCULAR ARC
10 L Y+0
11 DEP LT LEN10 ..... DEPART TANGENTIALLY (AUX.PT.)
12 L Z+100 R0 F MAX M2
13 END PGM 209 MM
```

Task: **Hexagon (polar)**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Call tool data	*TOOL CALL . . . . . S . . .*
Move to clearance height	*L Z . . .*
Define pole	*CC X . . . Y . . .*
Auxiliary point in polar coord.	*LP PR . . . PA . . .*
Plunging depth	*L Z . . .*
Approach tangentially	*APPR PLCT PR . . . PA . . . R . . .*
Contour	*LP PR . . . PA . . .*
⋮	⋮
Depart tangentially	*DEP PLCT PR . . . PA . . . R . . .*
Retract tool, PGM end	*L Z . . .*

Solution: **Hexagon (polar)**



Complete program

```
0 BEGIN PGM 213 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-20
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+0 Y+0 ..... POLE
6 LP PR+80 PA+0 R0 F MAX ..... AUXILIARY POINT (POLAR)
7 L Z-5 F MAX M3
8 APPR PLCT PR+40 PA+0 R5 RR F250 M8 ..... APPROACH STARTING POINT OF
9 LP PA+60 CONTOUR TANGENTIALLY
10 LP PA+120
11 LP PA+180
12 LP PA+240
13 LP PA+300
14 LP PA+360
15 DEP PLCT PR+80 PA+0 R5 ..... DEPART TANGENTIALLY
16 L Z+100 R0 F MAX M2
17 END PGM 213 MM
```

Task: **Circle (polar) CP**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Call tool data	*TOOL CALL . . . . . S . . .*
Move to clearance height	*L . . .*
Define pole	*. . .*
Auxiliary point in polar coord.	*. . .*
Contour	*. . .*
Auxiliary point	*. . .*
Retract tool, PGM end	*. . .*

Solution: **Circle (polar) CP**



Complete program

```
0 BEGIN PGM 211 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 CC X+50 Y+50 ..... POLE
6 LP PR+80 PA+180 R0 F MAX M3 ..... AUXILIARY POINT
7 L Z-5 F MAX
8 APPR PLCT PR+40 PA+180 R5 RL F250 M8 ..... APPROACH STARTING POINT OF
                                              CONTOUR TANGENTIALLY
9 CP IPA+360 DR- ..... CIRCLE PATH (POLAR)
10 DEP PLCT PR+80 PA+180 R5 ..... DEPART TANGENTIALLY
11 L Z+100 F MAX M2
12 END PGM 211 MM
```

Task: **Circular path with tangential connection (polar) CTP**

Program(s): _____



Begin program Define workpiece blank	*BEGIN PGM . . . MM* *BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Call tool data Move to clearance height	*TOOL CALL . . . . . S . . .* *L . . .*
Define pole Auxiliary point in polar coord.	*. . .*
Contour	*. . .*
Auxiliary point	*. . .*
Retract tool, PGM end	*. . .*
	. . .

Solution: **Circular path with tangential connection (polar) CTP**



Complete program

```
0 BEGIN PGM 212 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+70 Z+0
3 TOOL CALL 13 Z S4000 ..... R20
4 L Z+100 R0 F MAX
5 L X-30 Y-30 R0 F MAX M3 ..... AUXILIARY POINT
6 L Z-5 F MAX
7 APPR LT X+0 Y+0 LEN5 RL F250 M8 ..... APPROACH STARTING POINT OF
8 L X+0 Y+30 CONTOUR TANGENTIALLY
9 L X+40 Y+60
10 CC Y+20 X+20 ..... POLE
11 CTP PR+76,158 PA+23,199 ..... TANGENTIAL CIRCULAR PATH
                                 (POLAR)
12 L Y+0
13 DEP LT LEN5 ..... DEPART TANGENTIALLY
14 L Z+100 R0 F MAX M2
15 END PGM 212 MM
```

Task: **Polar coordinates (general)**

Program(s): _____



Solution: **Polar coordinates (general)**



Complete program

```
0 BEGIN PGM 252 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX
5 L X+130 Y+50 F MAX M3
6 L Z-5 F MAX
7 APPR LCT X+92 Y+50 R5 RL F250 M8 ..... APPROACH STARTING POINT OF
8 L Y+40 CONTOUR TANGENTIALLY
9 CC X+92 Y+40 ..... POLE
10 LP PR+45 PA-135 ..... STRAIGHT LINE (POLAR)
11 CC X+92 Y+50
12 LP PR+90 PA-150
13 CP PA+148 DR- ..... CIRCULAR PATH (POLAR)
14 L IX+30 IY+0
15 CC ..... DEFINE AS POLE POSITION
16 LP PR+21 PA-60
17 CC X+92 Y+88
18 CTP PR+29 PA-90 ..... TANGENTIAL PATH (POLAR)
19 L Y+50
20 DEP LCT X+130 Y+50 R5 ..... DEPART TANGENTIALLY
21 L Z+100 R0 F MAX M2
22 END PGM 252 MM
```

Task: **Drilling cycle**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*
Define workpiece blank	*BLK-FORM 0.1 . . . X . . . Y . . . Z . . .* *BLK-FORM 0.2 X . . . Y . . . Z . . .*
Call tool data	*TOOL CALL . . . . . S . . .*
Define cycle	*CYCL DEF . . .*
Move to clearance height	*L Z+ . . .*
Starting point first hole / call cycle	*L X . . . Y . . . M99*
Second hole / call cycle	*L X . . . Y . . . M99*
Third hole / call cycle	*L X . . . Y . . . M99*
Retract tool, PGM end	*. . .*

Solution: **Drilling cycle**



Complete program

```
0 BEGIN PGM 201 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1000 ..... R4
4 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-23 ..... DEPTH
  Q206=150 ..... FEED RATE FOR PLUNGING
  Q202=5 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
5 L Z+100 R0 F9999 M3
6 L X+20 Y+30 M99 ..... STARTING POSITION
7 L X+50 Y+70 M99 ..... 2ND HOLE
8 L X+75 Y+30 M99 ..... 3RD HOLE
9 L Z+100 R0 F MAX M2
10 END PGM 201 MM
```

Task: **Drilling with 200-series cycles** Program(s): _____



Procedure:

- Centering **Cycle 1**
- Drilling diameter 28 mm **Cycle 203**
- Boring **Cycle 202**
- Reaming **Cycle 201**
- Tapping **Cycle 2**



Complete program

```
0 BEGIN PGM 260 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-70
2 BLK FORM 0.2 X+50 Y+50 Z+0
```

Centering

```
3 TOOL CALL 9 Z S1000 ..... R6
4 L Z+100 R0 F MAX
5 CYCL DEF 1.0 PECKING
6 CYCL DEF 1.1 SET UP 2
7 CYCL DEF 1.2 DEPTH -5
8 CYCL DEF 1.3 PECKG 5
9 CYCL DEF 1.4 DWELL 0
10 CYCL DEF 1.5 F200
11 L X+0 Y+0 R0 F9999 M3
12 L Z+2 M99
13 L Z+100 M6
```

Drilling

```
14 TOOL CALL 11 Z S350 ..... R10
15 CYCL DEF 203 UNIVERSAL DRILLING
   Q200=5 ..... SET UP CLEARANCE
   Q201=-80 ..... DEPTH
   Q206=100 ..... FEED RATE FOR PLUNGING
   Q202=15 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q212=2 ..... DECREMENT
   Q213=2 ..... NR OF BREAKS
   Q205=5 ..... MIN. PLUNGING DEPTH
   Q211=0 ..... DWELL TIME AT DEPTH
   Q208=500 ..... RETRACTION FEED RATE
16 CYCL CALL M3
17 L Z+100 M6
```

Boring

```
18 TOOL CALL 13 Z S500 ..... R20
19 CYCL DEF 202 BORING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-35 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q211=0 ..... DWELL TIME AT DEPTH
   Q208=500 ..... RETRACTION FEED RATE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q214=0 ..... DISENGAGING DIRECTION
20 CYCL CALL M3
21 L Z+100 M6
```

Reaming

```
22 TOOL CALL 13 Z S100 ..... R20
23 CYCL DEF 201 REAMING
   Q200=5 ..... SET UP CLEARANCE
   Q201=-25 ..... DEPTH
   Q206=100 ..... FEED RATE FOR PLUNGING
   Q211=0 ..... DWELL TIME AT DEPTH
   Q208=300 ..... RETRACTION FEED RATE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
24 CYCL CALL M3
25 L Z+100 M6
```

Tapping

```
26 TOOL CALL 12 Z S50 ..... R15
27 CYCL DEF 2.0 TAPPING
28 CYCL DEF 2.1 SET UP 5
29 CYCL DEF 2.2 DEPTH -45
30 CYCL DEF 2.3 DWELL 0
31 CYCL DEF 2.4 F100
32 L Z-30 M3
33 CYCL CALL

34 L Z+100 M2
35 END PGM 260 MM
```

Task: **Slot plate**

Program(s): _____



Solution: **Slot plate**



Complete program

```
0 BEGIN PGM 210 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S1000 ..... R5
4 L Z+100 R0 F9999 M3
5 CYCL DEF 210 SLOT RECIP. PLNG
   Q200=2 ..... SET UP CLEARANCE
   Q201=-10 ..... DEPTH
   Q207=100 ..... FEED RATE FOR MILLNG
   Q202=5 ..... PLUNGING DEPTH
   Q215=0 ..... MACHINING OPERATION
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+25 ..... CENTER IN 2ND AXIS
   Q218=75 ..... FIRST SIDE LENGTH
   Q219=15 ..... SECOND SIDE LENGTH
   Q224=+20 ..... ANGLE OF ROTATION
6 CYCL CALL

7 CYCL DEF 211 CIRCULAR SLOT
   Q200=2 ..... SET UP CLEARANCE
   Q201=-10 ..... DEPTH
   Q207=250 ..... FEED RATE FOR MILLNG
   Q202=5 ..... PLUNGING DEPTH
   Q215=0 ..... MACHINING OPERATION
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND SET UP CLEARANCE
   Q216=+40 ..... CENTER IN 1ST AXIS
   Q217=+60 ..... CENTER IN 2ND AXIS
   Q244=50 ..... PITCH CIRCLE DIAMETER
   Q219=12 ..... SECOND SIDE LENGTH
   Q245=+35 ..... STARTING ANGLE
   Q248=180 ..... ANGULAR LENGTH
8 CYCL CALL

9 L Z+100 M2
10 END PGM 210 MM
```

Task: **Bushing plate**

Program(s): _____



Program layout: **Bushing plate**

Conventional preparation:	*BLK- FORM*	
First tool **Rough out circular pocket**	*TOOL CALL . . .* *CYCL DEF . . .* *. . . M99*	(Oversizing DR!)
Rough out first long slot	*CYCL DEF . . .* *CYCL CALL*	
Rough out second long slot	*FN 0: Q . . .* *CYCL CALL* *. . . M6*	Tool change
Second tool **Finish first short slot**	*TOOL CALL . . .* *CYCL DEF . . .* *CYCL CALL*	
Finish second short slot	*FN 0: Q . . .* *CYCL CALL*	
Finish first long slot	*CYCL DEF . . .* *CYCL CALL*	
Finish second long slot	*FN 0: Q . . .* *CYCL CALL*	
First rounded slot	*CYCL DEF . . .* *CYCL CALL*	
Second rounded slot	*FN 0: Q . . .* *CYCL CALL*	
Finish circular pocket	*CYCL DEF . . .* *CYCL CALL*	
Retract tool, end	*L Z100 M2*	

Solution: **Bushing plate**



Complete program

```
0  BEGIN PGM 262 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-15
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 6 Z S1500 DR+0,5 ..... R3
4  L Z+100 R0 F9999
5  L X+50 Y+50 M3
6  CYCL DEF 5.0 CIRCULAR POCKET
7  CYCL DEF 5.1 SET UP2
8  CYCL DEF 5.2 DEPTH -15
9  CYCL DEF 5.3 PECKG 5 F100
10 CYCL DEF 5.4 RADIUS 10
11 CYCL DEF 5.5 F200 DR+
12 L Z+2 M99
13 CYCL DEF 210 SLOT RECIP. PLNG
     Q200=2 ..... SET UP CLEARANCE
     Q201=-6 ..... DEPTH
     Q207=200 ..... FEEDRATE FOR MILNG
     Q202=6 ..... PLUNGING DEPTH
     Q215=0 ..... MACHINING OPERATION
     Q203=+0 ..... SURFACE COORDINATE
     Q204=20 ..... 2ND SET UP CLEARANCE
     Q216=+50 ..... CENTER IN 1ST AXIS
     Q217=+15 ..... CENTER IN 2ND AXIS
     Q218=81 ..... FIRST SIDE LENGTH
     Q219=11 ..... SECOND SIDE LENGTH
     Q224=+0 ..... ANGLE OF ROTATION
14 CYCL CALL
15 FN 0: Q217 = +85
16 CYCL CALL
17 L Z+100 M6
```

Solution: **Bushing plate**

```
18 TOOL CALL 5 Z S2000 ..... R2,5
19 CYCL DEF 210 SLOT RECIP. PLNG
   Q200=2 ..... SET UP CLEARANCE.
   Q201=-9 ..... DEPTH
   Q207=200 ..... FEEDRATE FOR MILNG
   Q202=5 ..... PLUNGING DEPTH
   Q215=0 ..... MACHINING OPERATION
   Q203=-6 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+15 ..... CENTER IN 2ND AXIS
   Q218=76,6 ..... FIRST SIDE LENGTH
   Q219=6,6 ..... SECOND SIDE LENGTH
   Q224=+0 ..... ANGLE OF ROTATION
20 CYCL CALL M3
21 FN 0: Q217 = +85
22 CYCL CALL
23 CYCL DEF 210 SLOT RECIP. PLNG
   Q200=2 ..... SET UP CLEARANCE
   Q201=-6,8 ..... DEPTH
   Q207=200 ..... FEED RATE FOR MILLNG
   Q202=6,8 ..... PLUNGING DEPTH
   Q215=0 ..... MACHINING OPERATION
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+15 ..... CENTER IN 2ND AXIS
   Q218=81 ..... FIRST SIDE LENGTH
   Q219=11 ..... SECOND SIDE LENGTH
   Q224=+0 ..... ANGLE OF ROTATION
24 CYCL CALL
25 FN 0: Q217 = +85
26 CYCL CALL
27 CYCL DEF 211 CIRCULAR SLOT
   Q200=2 ..... SET UP CLEARANCE
   Q201=-15 ..... DEPTH
   Q207=200 ..... FEEDRATE FOR MILLING
   Q202=5 ..... PLUNGING DEPTH
   Q215=0 ..... MACHINING OPERATION
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q244=60 ..... PITCH CIRCLE DIAMETER
   Q219=6,6 ..... SECOND SIDE LENGTH
   Q245=-30 ..... STARTING ANGLE
   Q248=60 ..... ANGULAR LENGTH
28 CYCL CALL
29 FN 0: Q245 = +150
30 CYCL CALL
31 CYCL DEF 214 C. POCKT FINSHNG
   Q200=2 ..... SET UP CLEARANCE
   Q201=-15 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=8 ..... PLUNGING DEPTH
   Q207=250 ..... FEEDRATE FOR MILLING
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q222=19 ..... WORKPIECE BLANK DIAMETER
   Q223=20,02 ..... FINISHED PART DIAMETER
32 CYCL CALL
33 L Z+100 R0 F MAX M2
34 END PGM 262 MM
```

Task: **Die I**

Program(s): _____



Solution: **Die I**



Complete program

```
0 BEGIN PGM 265 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 9 Z S500 DR+1 ..... R6
4 L Z+100 R0 F9999
5 CC X+50 Y+50
6 LP PR+70 PA+0
7 L Z+2 M3
8 L Z-10 F200
9 CP PA+360 DR-
10 LP PR+60
11 CP PA+360 DR-
12 LP PR+45 RL
13 CP PA+360 DR-
14 LP PR+65 PA+0 R0
15 L Z+2
16 CYCL DEF 4.0 POCKET MILLING
17 CYCL DEF 4.1 SET UP 2
18 CYCL DEF 4.2 DEPTH -5
19 CYCL DEF 4.3 PECKG 5 F100
20 CYCL DEF 4.4 X60
21 CYCL DEF 4.5 Y30
22 CYCL DEF 4.6 F250 DR- RADIUS 7
23 L X+50 Y+50 R0 F MAX M99
24 L Z+100 F MAX M6
```

Solution: **Die I**

```
25 TOOL CALL 8 Z S1000 ..... R5
26 CYCL DEF 212 POCKET FINISHING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-5 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=5 ..... PLUNGING DEPTH
   Q207=250 ..... FEED RATE FOR MILLNG
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q218=60 ..... FIRST SIDE LENGTH
   Q219=30 ..... SECOND SIDE LENGTH
   Q220=5,25 ..... CORNER RADIUS
   Q221=0 ..... ALLOWANCE IN 1ST AXS
27 CYCL CALL M3

28 CYCL DEF 215 C. STUD FINISHNG
   Q200=2 ..... SET UP CLEARANCE
   Q201=-10 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=10 ..... PLUNGING DEPTH
   Q207=250 ..... FEED RATE FOR MILLNG
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q222=92 ..... WORKPIECE BLANK DIAMETER
   Q223=90 ..... FINISHED PART DIAMETER
29 CYCL CALL

30 L Z+100 R0 F MAX M2
31 END PGM 265 MM
```

Task: **Linear hole pattern**

Program(s): _____





Solution: **Linear hole pattern**



Main program

```
0 BEGIN PGM 220 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-20 ..... DEPTH
  Q206=100 ..... FEED RATE FOR PLUNGING
  Q202=5 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
6 CALL LBL 1
```

Retract tool, end

```
7 L Z+100 M2
```

SPGM

```
8 LBL 1
9 CYCL DEF 221 CARTESIAN PATTRN
  Q225=+15 ..... STARTNG PNT 1ST AXIS
  Q226=+20 ..... STARTNG PNT 2ST AXIS
  Q237=+10 ..... SPACING IN 1ST AXIS
  Q238=+8 ..... SPACING IN 2ND AXIS
  Q242=8 ..... NUMBER OF COLUMNS
  Q243=9 ..... NUMBER OF ROWS
  Q224=+0 ..... ANGLE OF ROTATION
  Q200=2 ..... SET UP CLEARANCE
  Q203=+0 ..... SURFACE COORDINATE
  Q204=2 ..... 2ND SET UP CLEARANCE
10 LBL 0
11 END PGM 220 MM
```

Task: **Hole pattern**

Program(s): _____



Solution: **Hole pattern**



Main program

```
0 BEGIN PGM 221 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 200 DRILLING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-17 ..... DEPTH
   Q206=150 ..... FEED RATE FOR PLUNGING
   Q202=6 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND SET UP CLEARANCE
6 CALL LBL 1
7 CALL LBL 2
```

Retract tool, end

```
8 L Z+100 M2
```

SPGM

```
9 LBL 1
10 CYCL DEF 220 POLAR PATTERN
   Q216=+90 ..... CENTER IN 1ST AXIS
   Q217=+10 ..... CENTER IN 2ND AXIS
   Q244=100 ..... PITCH CIRCLE DIA.
   Q245=+90 ..... STARTING ANGLE
   Q246=+180 ..... STOPPING ANGLE
   Q247=15 ..... STEPPING ANGLE
   Q241=7 ..... NR OF REPETITIONS
   Q200=2 ..... SET UP CLEARANCE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND SET UP CLEARANCE
11 LBL 0
```

Solution: **Hole pattern**

```
12 LBL 2
13 CYCL DEF 221 CARTESIAN PATTRN
   Q225=+30 ..... STARTNG PNT 1ST AXIS
   Q226=+30 ..... STARTNG PNT 2ND AXIS
   Q237=+15 ..... SPACING IN 1ST AXIS
   Q238=+20 ..... SPACING IN 2ND AXIS
   Q242=5 ..... NUMBER OF COLUMNS
   Q243=2 ..... NUMBER OF ROWS
   Q224=+80 ..... ANGLE OF ROTATION
   Q200=2 ..... SET UP CLEARANCE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND  SET UP CLEARANCE
14 LBL 0
15 END PGM 221 MM
```

Task: **Guide plate**

Program(s): _____



Procedure:

• Shoulders	
• Centering	**Cycle 200**
• Drilling	**Cycle 200**
• Tapping	**Cycle 2**

Solution: **Guide plate**



Main program

```
0  BEGIN PGM 261 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-30
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 13 Z S500 ..... R20
4  L Z+100 R0 F9999
5  L X-50 Y-50
6  L Z+2 M3
7  L Z-10 F200
8  APPR LCT X+30 Y+0 R2 RL
9  L Y+100
10 DEP LCT X+0 Y+150 R2
11 L X+100 F MAX
12 APPR LCT X+70 Y+100 R2 RL
13 L Y+0
14 DEP LCT X+100 Y-50 R2
15 L Z+100 R0 F9999 M6

16 TOOL CALL 4 Z S2000 ..... R2,5
17 CYCL DEF 200 DRILLING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-3,5 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=3,5 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=-10 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
18 L Z+5 M3
19 CALL LBL 1
20 L Z+100 M6
```

Solution: **Guide plate**

```
21 TOOL CALL 5 Z S2000 ..... R2,5
22 CYCL DEF 200 DRILLING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-15 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=5 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=-10 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
23 L Z+5 M3
24 CALL LBL 1
25 L Z+100 M6

26 TOOL CALL 6 Z S300 ..... R3
27 CYCL DEF 2.0 TAPPING
28 CYCL DEF 2.1 SET UP 2
29 CYCL DEF 2.2 DEPTH -12
30 CYCL DEF 2.3 DWELL 0
31 CYCL DEF 2.4 F300
32 L Z+5 M3
33 CALL LBL 1
```

Retract tool, end

```
34 L Z+100 R0 F MAX M2
```

SPGM

```
35 LBL 1
36 CYCL DEF 221 CARTESIAN PATTRN
   Q225=+15 ..... STARTNG PNT 1ST AXIS
   Q226=+10 ..... STARTNG PNT 2ND AXIS
   Q237=+70 ..... SPACING IN 1ST AXIS
   Q238=+20 ..... SPACING IN 2ND AXIS
   Q242=2 ..... NUMBER OF COLUMNS
   Q243=5 ..... NUMBER OF LINES
   Q224=+0 ..... ANGLE OF ROTATION
   Q200=2 ..... SET UP CLEARANCE
   Q203=-10 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
```

SPGM, end

```
37 LBL 0
38 END PGM 261 MM
```

Task: **Die II**

Program(s): _____



Program layout: **Die II**

Define workpc. blank	*BLK FORM*	
	CALL PGM . . .	PGM-Call
Drilling ⌀ 5 mm	*CYCL DEF . . .*	
	CALL LBL . . .	SPGM1-Call
	L Z100 . . . M6	Tool change
Centering, pitch circle	*CYCL DEF . . .*	
	CALL LBL . . .	SPGM2-Call
	L Z100 . . . M6	Tool change
Drilling, pitch circle	*CYCL DEF . . .*	
	CALL LBL . . .	SPGM2-Call
	L Z100 . . . M6	Tool change
Tapping	*CYCL DEF . . .*	
	CALL LBL . . .	SPGM2-Call
Retract tool, end	*L Z . . . M2*	

SPGM1	*LBL 1*
Cartesian pattern	*CYCL DEF . . .*
	LBL 0
SPGM2	*LBL 2*
Polar pattern	*CYCL DEF . . .*
	LBL 0

Copy help program from PGM 265 

Solution: **Die II**



Main program

```
0 BEGIN PGM 266 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 CALL PGM 26501 ..... HELP PROGRAM
4 TOOL CALL 5 Z S2000 ..... R2,5
5 L Z+100 M3
6 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-6 ..... DEPTH
  Q206=200 ..... FEED RATE FOR PLUNGING
  Q202=6 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=-5 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE
7 CALL LBL 1
8 L Z+100 R0 F9999 M6

9 TOOL CALL 4 Z S2000 ..... R2
10 L Z+100 M3
11 CYCL DEF 1.0 PECKING
12 CYCL DEF 1.1 SET UP 2
13 CYCL DEF 1.2 DEPTH -2,8
14 CYCL DEF 1.3 PECKG 2,8
15 CYCL DEF 1.4 DWELL 0
16 CYCL DEF 1.5 F200
17 CALL LBL 2
18 L Z+100 M6

19 TOOL CALL 5 Z S1500 ..... R2,5
20 L Z+100 M3
21 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-15 ..... DEPTH
  Q206=200 ..... FEED RATE FOR PLUNGING
  Q202=8 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE
22 CALL LBL 2
23 L Z+100 M6
```

```
24 TOOL CALL 6 Z S300 ..... R3
25 L Z+100 M3
26 CYCL DEF 2.0 TAPPING
27 CYCL DEF 2.1 SET UP 2
28 CYCL DEF 2.2 DEPTH -12
29 CYCL DEF 2.3 DWELL 1
30 CYCL DEF 2.4 F300
31 CALL LBL 2

32 L Z+100 R0 F MAX M2
```

SPGM

```
33 LBL 1
34 CYCL DEF 221 CARTESIAN PATTRN
   Q225=+30 ..... STARTNG PNT 1ST AXIS
   Q226=+42 ..... STARTNG PNT 2ND AXIS
   Q237=+10 ..... SPACING IN 1ST AXIS
   Q238=+8 ..... SPACING IN 2ND AXIS
   Q242=5 ..... NUMBER OF COLUMNS
   Q243=3 ..... NUMBER OF LINES
   Q224=+0 ..... ANGLE OF ROTATION
   Q200=2 ..... SET UP CLEARANCE
   Q203=-6 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
35 LBL 0

36 LBL 2
37 CYCL DEF 220 POLAR PATTERN
   Q216=+50 ..... CENTER IN 1ST AXIS
   Q217=+50 ..... CENTER IN 2ND AXIS
   Q244=80 ..... PITCH CIRCLE DIA.
   Q245=+0 ..... STARTING ANGLE
   Q246=+360 ..... STOPPING ANGLE
   Q247=30 ..... STEPPING ANGLE
   Q241=12 ..... NR OF REPETITIONS
   Q200=2 ..... SET UP CLEARANCE
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
38 LBL 0
39 END PGM 266 MM
```

Task: **Hexagon**

Program(s): _____



Solution: **Hexagon**



Complete program

```
0  BEGIN PGM 268 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 12 Z S500 ..... R15
4  L Z+100 R0 F9999
5  CC X+50 Y+50 ..... POLE
6  LP PR+80 PA+0 R0 ..... AUXILIARY POINT
7  L Z-5 M3
8  APPR PLCT PR+40 PA+0 R2 RL F100 ..... STARTING POINT

9  LBL 1 ..... DEFINE LABEL
10 LP PR+40 IPA-60
11 CALL LBL 1 REP 5/5 ..... CALL LABEL WITH REPEATS

12 DEP PLCT PR+80 PA+0 R2 F200 ..... AUXILIARY POINT
13 L Z+100 R0 F MAX M2
14 END PGM 268 MM
```

Task: **Drilled plate – slanted columns**

Program(s): _____



Program layout: **Drilled plate – slanted columns**

Move up and down columns (meandering)



Drilling pattern

L X . . . Y . . . *L Z . . .*	Absolute pre-position Set up clearance
LBL 1	Define label 1
L IX . . . M99	Cross over to first hole
LBL 2	Define label 2
L IY . . . M99	Move up the column
CALL LBL 2 REP . . .	
L IX . . . M99	Cross over two second column
LBL 3	Define label 2
L IY . . . M99	Move down the column
CALL LBL 3 REP . . .	
CALL LBL 1 REP . . .	Remaining groups

Solution: **Drilled plate – slanted columns**



Main program

```
0 BEGIN PGM 270 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 L Z+100 R0 F MAX

4 TOOL CALL 5 Z S4000 ..... R2,5
5 CYCL DEF 200 DRILLING
  Q200=2 ..... SET UP CLEARANCE
  Q201=-15 ..... DEPTH
  Q206=250 ..... FEED RATE FOR PLUNGING
  Q202=8 ..... PLUNGING DEPTH
  Q210=0 ..... DWELL TIME AT TOP
  Q203=+0 ..... SURFACE COORDINATE
  Q204=20 ..... 2ND SET UP CLEARANCE

6 L X-10 Y+10 R0 F9999 M3
7 L Z+2

8 LBL 1
9 L IX+20 M99

10 LBL 2
11 L IX+3 IY+15 M99
12 CALL LBL 2 REP 4/4

13 L IX+10 M99

14 LBL 3
15 L IX-3 IY-15 M99
16 CALL LBL 3 REP 4/4

17 CALL LBL 1 REP 2/2

18 L Z+100 R0 F MAX M2
19 END PGM 270 MM
```

Task: **FK Cam**

Program(s): _____



Solution: **FK Cam**



Complete program

```
0  BEGIN PGM 288 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 10 Z S4000 ..... R7,5
4  L Z+100 R0 F MAX
5  L X-15 Y+35 R0 F MAX M3
6  L Z-5 R0 F MAX
7  APPR LCT X+5 Y+35 R5 RL F400
8  FC DR- R25 CCX+30 CCY+35
9  FLT
10 FCT DR- R15 CCX+50 CCY+75
11 FLT
12 FCT DR- R20 CCX+75 CCY+25
13 FLT
14 FCT X+5 Y+35 DR- R25 CCX+30 CCY+35
15 DEP LCT X-15 Y+35 R5
16 L Z+100 R0 F MAX M2
17 END PGM 288 MM
```

Task: **Maltese cross**

Program(s): _____



Solution: **Maltese cross**



Main program

```
0  BEGIN PGM 275 MM
1  BLK FORM 0.1 Z X-50 Y-50 Z-20
2  BLK FORM 0.2 X+50 Y+50 Z+0
3  TOOL CALL 8 Z S250074 ..... R5
4  L Z+100 R0 F MAX
5  L Z+2 R0 F MAX M3
6  CALL LBL 10

7  LBL 5
8  CYCL DEF 10.0 ROTATION
9  CYCL DEF 10.1 IROT-90
10 CALL LBL 10
11 CALL LBL 5 REP 2/2

12 CYCL DEF 10.0 ROTATION
13 CYCL DEF 10.1 ROT+0

14 L Z+100 R0 F MAX M2
```

SPGM

```
15 LBL 10
16 CC X+0 Y+0
17 LP PR+70 PA+180 R0 F1000
18 L Z-10
19 APPR PLCT PR+50 PA+180 R2 RL F500
20 FPOL X+0 Y+0
21 FC DR- R50 CCX+0 CCY+0
22 FL AN+65 PDX+0 PDY+0 D48
23 FSELECT 4
24 FC DR+ R38 CCPR+75 CCPA+135
25 FSELECT 2
26 FL AN+25 PDX+0 PDY+0 D48
27 FSELECT 1
28 FC PR+50 PA+90 DR- R50 CCA+90
29 DEP PLCT PR+70 PA+90 R2
30 LBL 0
31 END PGM 275 MM
```

Task: **FK Hammer**

Program(s): _____





Detail Z



Solution: **FK Hammer**



Complete program

```
0  BEGIN PGM 289 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 11 Z S4000
4  L Z+100 R0 F MAX
5  CC X+30 Y+40
6  LP PR+60 PA+30 R0 F MAX M3
7  L Z+2 R0 F MAX
8  L Z-5 R0 F100
9  APPR PLCT PR+37,5 PA+30 R5 RL F400
10 FL LEN10 AN-60
11 FL AN-150
12 FCT DR+ R12,5
13 FLT AN-100
14 FCT DR- R11 CCX+30 CCY+15
15 FLT
16 FCT DR- R11 CCX+15 CCY+65
17 FLT AN-55
18 FCT DR+ R15
19 FLT AN+30
20 FPOL X+30 Y+40
21 FL PR+37,5 PA+30 LEN10 AN-60
22 DEP PLCT PR+60 PA+30 R5
23 L Z+100 R0 F MAX M2
24 END PGM 289 MM
```

Task: **FK Hook-type spanner wrench** Program(s): _____



Solution:

FK Hook-type spanner wrench



Complete program

```
0 BEGIN PGM 295 MM
1 BLK FORM 0.1 Z X+0 Y-35 Z-20
2 BLK FORM 0.2 X+160 Y+35 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX
5 L X-10 Y-10 R0 F MAX
6 L Z-5 R0 F MAX M3
7 APPR LCT X+0 Y+0 R2 RR F400
8 FL AN+0
9 FCT DR- R70
10 FCT DR+ R60 CCX+105 CCY+28
11 FSELECT 2
12 FCT DR+ R12
13 FCT DR+ R18 CCX+138 CCY-5
14 FLT LEN5 AN+170
15 FL X+134 AN-150
16 FSELECT 1
17 FL LEN4 IAN-90
18 FCT DR- R8
19 FCT DR- R60 CCX+105
20 FSELECT 1
21 FCT DR- R40 CCX+100
22 FCT  Y+20 DR+ R5
23 FSELECT 2
24 FLT X+0 AN+180
25 FSELECT 1
26 L Y+0
27 DEP LCT X-10 Y-10 R2
28 L Z+100 R0 FMAX M2
29 END PGM 295 MM
```

Task: **Subprogram (Groups of holes)**

Program(s): _____



Begin program	*BEGIN PGM . . . MM*	
	• •	
	TOOL CALL . . .	Call tool data
	CYCL DEF . . .	Define cycle
	L Z+ . . .	Move to clear. height
	L X . . . Y . . .	Start. pos.-hole group ①
	L Z . . .	Set up clearance
	CALL LBL . . .	Call SPGM
	L X . . . Y . . .	Start. pos.-hole group ②
	CALL LBL . . .	Call SPGM
	L X . . . Y . . .	Start. pos.-hole group ③
	CALL LBL . . .	Call SPGM
Retract tool, end	*L Z +100 R0 F9999 M2*	

SPGM	*LBL . . .*
SPGM end	*LBL 0*

Solution: **Subprogram (Groups of holes)**



Main program

```
0  BEGIN PGM 215 MM
1  BLK FORM 0.1 Z X+0 Y-70 Z-20
2  BLK FORM 0.2 X+100 Y+0 Z+0
3  TOOL CALL 4 Z S4000 ..... CALL TOOL DATA,  R2
4  CYCL DEF 200 DRILLING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-18 ..... DEPTH
   Q206=200 ..... FEED RATE FOR PLUNGING
   Q202=4 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=+0 ..... SURFACE COORDINATE
   Q204=2 ..... 2ND SET UP CLEARANCE
5  L Z+100 R0 F MAX ..... CLEARANCE HEIGHT
6  L X+20 Y-10 R0 F MAX ..... STARTING POINT-HOLE GROUP ①
7  L Z+2 R0 F MAX M13
8  CALL LBL 1 ..... CALL SPGM

9  L X+63 Y-12 F MAX ..... STARTING POINT-HOLE GROUP ②
10 CALL LBL 1 ..... CALL SPGM

11 L X+42 Y-32 F MAX ..... STARTING POINT-HOLE GROUP ③
12 CALL LBL 1 ..... CALL SPGM
```

Retract tool, end

```
13 L Z+100 R0 F MAX M2
```

SPGM

```
14 LBL 1
15 CYCL CALL
16 L IX+15 F MAX M99
17 L IY-15 F MAX M99 ..... DRILLING PATTERN
18 L IX-15 F MAX M99
```

SPGM, end

```
19 LBL 0
20 END PGM 215 MM
```

Task: **Bolt hole circle segments with several tools**

Program(s): _____



Program layout: **Nesting for double bolt hole circles**

Conventional Preparation:	*BLK- FORM*	
Centering	*TOOL CALL . . .* *CYCL DEF / L Z+100*	
	CALL LBL 1	
Drilling	*TOOL CALL . . .* *CYCL DEF . . .*	
	CALL LBL 1	
Tapping	*TOOL CALL . . .* *CYCL DEF . . .*	
	CALL LBL 1	
Retract tool, end	*L Z100 M2*	
SPGM1	*LBL 1*	
	CC X . . . Y . . .	Circle center-left
	CALL LBL 2	Call bolt hole circle seg.
	CC X . . . Y . . .	Circle center-right
	CALL LBL 2	Call bolt hole circle seg.
	• • •	
SPGM1 end	*LBL 0*	
SPGM2, Bolt hole circle segments	*LBL 2*	
	LP PR . . . PA . . . M3 *L Z+2 M99*	Starting position
	LBL 3	Remaining holes
	• • •	
	CALL LBL 3 REP . . .	
Partial repeat	*LP PR . . . PA . . .*	
	LBL 4	
	• • •	
	CALL LBL 4 REP . . .	
SPGM2 end	*LBL 0*	

Solution: **Bolt hole circle segments with several tools**



Main program

```
0  BEGIN PGM 280 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+150 Y+60 Z+0
3  TOOL CALL 4 Z S2000 ..... R2
4  L Z+100 R0 F9999
5  CYCL DEF 1.0 PECKING
6  CYCL DEF 1.1 SET UP 2
7  CYCL DEF 1.2 DEPTH -3,5
8  CYCL DEF 1.3 PECKG 3,5
9  CYCL DEF 1.4 DWELL 0
10 CYCL DEF 1.5 F200
11 CALL LBL 1
12 L Z+100 M6

13 TOOL CALL 5 Z S1500 ..... R2,5
14 CYCL DEF 200 DRILLING
   Q200=2 ..... SET UP CLEARANCE
   Q201=-15 ..... DEPTH
   Q206=250 ..... FEED RATE FOR PLUNGING
   Q202=8 ..... PLUNGING DEPTH
   Q210=0 ..... DWELL TIME AT TOP
   Q203=+0 ..... SURFACE COORDINATE
   Q204=20 ..... 2ND SET UP CLEARANCE
15 CALL LBL 1
16 L Z+100 M6

17 TOOL CALL 6 Z S300 ..... R3
18 CYCL DEF 2.0 TAPPING
19 CYCL DEF 2.1 SET UP 2
20 CYCL DEF 2.2 DEPTH -12
21 CYCL DEF 2.3 DWELL  1
22 CYCL DEF 2.4 F300
23 CALL LBL 1

24 L Z+100 R0 M2
```

Solution: **Bolt hole circle segments with several tools**

SPGM

```
25 LBL 1
26 CC X+40 Y+0
27 CALL LBL 2
28 CC X+110 Y+0
29 CALL LBL 2
30 LBL 0

31 LBL 2
32 LP PR+55 PA+120 R0 M3
33 L Z+2 M99

34 LBL 3
35 LP IPR-10 M99
36 CALL LBL 3 REP 3/3

37 LP PR+15 PA+60 R0 M99

38 LBL 4
39 LP IPR+10 M99
40 CALL LBL 4 REP 3/3

41 LBL 0
42 END PGM 280 MM
```

Task: **Milling with several settings**

Program(s): _____



Solution: **Milling with several settings**



Main program

```
0 BEGIN PGM 223 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-40
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 13 Z S2500 ..... R20
4 L Z+100 R0 F MAX M3
5 L X-30 Y+70 R0 F MAX ..... STARTING POSITION
6 L Z+0 F MAX

7 LBL 2
8 L IZ-5 R0 F MAX M3 ..... SETTING
9 CALL LBL 1 ..... CALL CONTOUR

10 CALL LBL 2 REP 5/5 ..... FURTHER CONTOUR SECTIONS
```

Retract tool, end

```
11 L Z+100 R0 F MAX M2
```

SPGM, Contour

```
12 LBL 1
13 APPR LCT X+10 Y+70 R5 RL F250 M3
14 L X+10 Y+90 RL
15 RND R10
16 L X+50 Y+90
17 RND R20
18 L X+90 Y+50
19 RND R20 ..... CONTOUR
20 L X+90 Y+10
21 RND R10
22 L X+50 Y+10
23 RND R20
24 L X+10 Y+50
25 RND R20
26 L X+10 Y+70
27 DEP LCT X-20 Y+70 R5 F500
28 LBL 0
```

SPGM end

```
29 END PGM 223 MM
```

Task: **Contour cycles SL II, Roughing out kidney**

Program(s): _____



Solution: **Contour cycles SL II, Roughing out kidney**



Main program

```
0  BEGIN PGM 240 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 7 Z S1500 ..... R4
4  L Z+100 R0 F9999 M3
5  CYCL DEF 14.0 CONTOUR GEOMETRY
6  CYCL DEF 14.1 CONTOUR LABEL 1
7  CYCL DEF 20.0 CONTOUR DATA
     Q1=-18 ..... MILLING DEPTH
     Q2=1 ..... TOOL PATH OVERLAP
     Q3=+0,5 ..... ALLOWANCE FOR SIDE
     Q4=+0,5 ..... ALLOWANCE FOR FLOOR
     Q5=+0 ..... WORKPIECE SURFACE COORD.
     Q6=2 ..... SET UP CLEARANCE
     Q7=+10 ..... CLEARANCE HEIGHT
     Q8=0 ..... ROUNDING RADIUS
     Q9=-1 ..... DIRECTION OF ROTATION
8  CYCL DEF 22.0 ROUGH OUT
     Q10=10 ..... PLUNGING DEPTH
     Q11=100 ..... FEED RATE FOR PLUNGING
     Q12=200 ..... FEED RATE FOR MILLING
     Q18=0 ..... COARSE ROUGHING TOOL
     Q19=150 ..... RECIPROCATION FEED RATE
9  CYCL CALL
10 L Z+100 R0 F MAX M6

11 TOOL CALL 8 Z S2000 ..... R5
12 CYCL DEF 23.0 FLOOR FINISHING
     Q11=100 ..... FEED RATE FOR PLUNGING
     Q12=200 ..... FEED RATE FOR MILLING
13 CYCL CALL M3
14 CYCL DEF 24.0 SIDE FINISHING
     Q9=-1 ..... DIRECTION OF ROTATION
     Q10=1 ..... PLUNGING DEPTH
     Q11=100 ..... FEED RATE FOR PLUNGING
     Q12=200 ..... FEED RATE FOR MILLING
     Q14=+0 ..... ALLOWANCE FOR SIDE
15 CYCL CALL M3
```

Retract tool, end

```
16 L Z+100 R0 F MAX M2
```

SPGM

```
17 LBL 1
18 L X+5 Y+30 RR
19 CC X+15 Y+30
20 C X+6,645 Y+35,495 DR-
21 CT X+55,505 Y+69,488
22 CT X+58,995 Y+30,025
23 CT X+19,732 Y+21,191
24 C X+5 Y+30 DR-
25 LBL 0
26 END PGM 240 MM
```

Task: **Contour cycles SL II, Kidney shaped island**

Program(s): _____



Solution: **Contour cycles SL II, Kidney shaped island**



Main program

```
0 BEGIN PGM 241 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 7 Z S1500 ..... R3
4 L Z+100 R0 F9999 M3
5 CYCL DEF 14.0 CONTOUR GEOMETRY
6 CYCL DEF 14.1 CONTOUR LABEL 1 /2
7 CYCL DEF 20.0 CONTOUR DATA
   Q1=-18 ..... MILLING DEPTH
   Q2=1 ..... TOOL PATH OVERLAP
   Q3=+0,5 ..... ALLOWANCE FOR SIDE
   Q4=+0,5 ..... ALLOWANCE FOR FLOOR
   Q5=+0 ..... WORKPIECE SURFACE COORD.
   Q6=2 ..... SET UP CLEARANCE
   Q7=+10 ..... CLEARANCE HEIGHT
   Q8=0 ..... ROUNDING RADIUS
   Q9=-1 ..... DIRECTION OF ROTATION
8 CYCL DEF 22.0 ROUGH OUT
   Q10=10 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q18=0 ..... COARSE ROUGHING TOOL
   Q19=150 ..... RECIPROCATION FEED RATE
9 CYCL CALL
10 L Z+100 R0 F MAX M6

11 TOOL CALL 8 Z S2000 ..... R4
12 CYCL DEF 23.0 FLOOR FINISHING
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
13 CYCL CALL M3
14 CYCL DEF 24.0 SIDE FINISHING
   Q9=-1 ..... DIRECTION OF ROTATION
   Q10=1 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q14=+0 ..... ALLOWANCE FOR SIDE
15 CYCL CALL M3
```

Retract tool, end

```
16 L Z+100 R0 F MAX M2
```

SPGM

```
17 LBL 1 ..... ISLAND
18 L X+5 Y+30 RL
19 CC X+15 Y+30
20 C X+6,645 Y+35,495 DR-
21 CT X+55,505 Y+69,488
22 CT X+58,995 Y+30,025
23 CT X+19,732 Y+21,191
24 C X+5 Y+30 DR-
25 LBL 0

26 LBL 2 ..... CHAMBER
27 L X-10 Y-10 RR
28 L Y+110
29 L X+110
30 L Y-10
31 L X-10
32 LBL 0
33 END PGM 241 MM
```

Task: **FK Mickey Mouse**

Program(s): _____



Detail Z



Solution: **FK Mickey Mouse with 20-series cycles**



Main program

```
0 BEGIN PGM 290 MM
1 ..... MICKEY WITH 20-SERIES CYCLES
2 ..... B06
3 BLK FORM 0.1 Z X-70 Y-90 Z-12
4 BLK FORM 0.2 X+70 Y+30 Z+0
5 TOOL CALL 7 Z S2500 ..... R4
6 L Z+20 R0 F9999 M3
7 CYCL DEF 14.0 CONTOR GEOMETRY
8 CYCL DEF 14.1 CONTOUR LABEL 1 /2
9 CYCL DEF 20.0 CONTOR DATA
   Q1=-10 ..... MILLING DEPTH
   Q2=1 ..... TOOL PATH OVERLAP
   Q3=+1 ..... ALLOWANCE FOR SIDE
   Q4=+1 ..... ALLOWANCE FOR FLOOR
   Q5=+0 ..... WORKPIECE SURFACE COORD.
   Q6=-2 ..... SET UP CLEARANCE
   Q7=+20 ..... CLEARANCE HEIGHT
   Q8=+0 ..... ROUNDING RADIUS
   Q9=-1 ..... DIRECTION OF ROTATION
10 CYCL DEF 22.0 ROUGH OUT
   Q10=-5 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q18=0 ..... COARSE ROUGHING TOOL
   Q19=150 ..... RECIPROCATION FEED RATE
11 L X-65 Y+0 M99

12 CYCL DEF 23.0 FLOOR FINISHING
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
13 CYCL CALL
14 YCL DEF 24.0 SIDE FINISHING
   Q9=-1 ..... DIRECTION OF ROTATION
   Q10=-5 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q14=+0 ..... ALLOWANCE FOR SIDE
15 CYCL CALL
```

Retract tool, end

```
16 L Z+100 R0 F MAX M2
```

SPGM

```
17 LBL 1
18 L X-62 Y+0 RL
19 FC DR- R22 CLSD+ CCX-40 CCY+0
20 FCT DR+ R38 CCX+0
21 FSELECT 1
22 FCT DR- R22 CCX+40 CCY+0
23 FCT DR+ R10
24 FLT AN-60 PDX+0 PDY+0 D0
25 FSELECT 3
26 FCT DR- R8
27 FCT X+0 Y-80 DR- R80 CCX+0 CCY+0
28 FLT X-40 AN+180
29 FL AN+90
30 FL AN+30
31 FCT DR+ R9 CCX-17
32 FLT X-30 PAR31 DP18
33 FSELECT 2
34 FL AN+90
35 FL PAR33 DP8
36 FSELECT 2
37 FCT DR+ R9 CCX-17
38 FCT DR- R22 CLSD- CCX-40 CCY+0
39 FSELECT 2
40 LBL 0

41 LBL 2
42 L X-74 Y+20 RR
43 L Y+35
44 L X+74
45 L Y-100
46 L X-74
47 L Y+0
48 LBL 0
49 END PGM 290 MM
```

Task: **FK-SL Combination**

Program(s): _____



Solution: **FK-SL Combination**



Main program

```
0  BEGIN PGM 273 MM
1  BLK FORM 0.1 Z X-50 Y-50 Z-30
2  BLK FORM 0.2 X+50 Y+50 Z+0
3  TOOL CALL 6 Z S1500 ..... R3
4  CYCL DEF 14.0 CONTOUR GEOMETRY
5  CYCL DEF 14.1 CONTOUR  LABEL 1 /2
6  CYCL DEF 20.0 CONTOUR DATA
   Q1=-10 ..... MILLING DEPTH
   Q2=1 ..... TOOL PATH OVERLAP
   Q3=+1 ..... ALLOWANCE FOR SIDE
   Q4=+1 ..... ALLOWANCE FOR FLOOR
   Q5=+0 ..... WORKPIECE SURFACE COORD.
   Q6=2 ..... SET UP CLEARANCE
   Q7=+20 ..... CLEARANCE HEIGHT
   Q8=0 ..... ROUNDING RADIUS
   Q9=-1 ..... DIRECTION OF ROTATION
7  CALL LBL 10

8  CYCL DEF 14.0 CONTOUR GEOMETRY
9  CYCL DEF 14.1 CONTOUR   LABEL 1 /3
10 FN 0: Q5 = -10
11 CALL LBL 10
12 L Z+100 R0 F MAX M6

13 TOOL CALL 4 Z S2000 ..... R2
14 CYCL DEF 14.0 CONTOUR GEOMETRY
15 CYCL DEF 14.1 CONTOUR  LABEL 1 /3
16 CALL LBL 11

17 CYCL DEF 14.0 CONTOUR GEOMETRY
18 CYCL DEF 14.1 CONTOUR  LABEL 1 /2
19 FN 0: Q5 = +0
20 CALL LBL 11
```

Retract tool, end

```
21 L Z+100 R0 F MAX M2
```

Solution: **FK-SL Combination**

SPGM

```
22 LBL 10
23 CYCL DEF 22.0 ROUGH OUT
   Q10=10 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q18=0 ..... COARSE ROUGHING TOOL
   Q19=150 ..... RECIPROCATION FEED RATE
24 CYCL CALL M3
25 LBL 0

26 LBL 11
27 CYCL DEF 23.0 FLOOR FINISHING
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
28 CYCL CALL M3
29 CYCL DEF 24.0 SIDE FINISHING
   Q9=-1 ..... DIRECTION OF ROTATION
   Q10=5 ..... PLUNGING DEPTH
   Q11=100 ..... FEED RATE FOR PLUNGING
   Q12=200 ..... FEED RATE FOR MILLING
   Q14=+0 ..... ALLOWANCE FOR SIDE
30 CYCL CALL
31 LBL 0

32 LBL 1
33 L X-40 Y+0 RR
34 L Y+40
35 RND R5
36 L X+40
37 RND R5
38 L Y-40
39 RND R5
40 L X-40
41 RND R5
42 L Y+0
43 LBL 0

44 LBL 2
45 CC X+0 Y+0
46 FPOL X+0 Y+0
47 LP PR+22 PA+45 RL
48 FC DR- R4 CLSD+
49 FLT AN-91,85
50 FCT DR- R16 CCX+0 CCY+0
51 FSELECT 1
52 FLT AN-178,15
53 FCT DR- R4
54 FLT AN+88,15
55 FCT DR- R16 CCX+0 CCY+0
56 FLT AN+1,85
57 FCT PR+22 PA+45 DR- R4 CLSD-
58 FSELECT 2
59 LBL 0
```

Solution: **FK-SL Combination**

```
60 LBL 3
61 CC X+0 Y+0
62 LP PR+28 PA+180 RL
63 FC Y+14 DR- R28 CCX+0 CCY+0
64 FSELECT 2
65 FL AN+0
66 FCT X-14 DR+ R5
67 FLT AN+90
68 FC X+14 DR- R28 CCX+0 CCY+0
69 FSELECT 1
70 FL AN-90
71 FCT Y+14 DR+ R5
72 FLT AN+0
73 FC Y-14 DR- R28 CCX+0 CCY+0
74 FSELECT 1
75 FL AN-180
76 FCT X+14 DR+ R5
77 FLT AN-90
78 FC X-14 DR- R28 CCX+0 CCY+0
79 FSELECT 2
80 FL AN+90
81 FCT Y-14 DR+ R5
82 FLT AN+180
83 FC X-28 Y+0 DR- R28 CCX+0 CCY+0
84 LBL 0
85 END PGM 273 MM
```

Task: **DEMO Tree**

Program(s): _____

_____





P0 ≠ Contour point

Point	X	Y	Point	X	Y
P0	0,0	0,0	M1	-5,5	13,5
P1	–	9,0	M2	2,5	24,0
P2	–	10,0	M3	12,0	13,0
			M4	–	-2,0
			M5	-3,0	-18,0
			M6	8,0	-18,0

Program layout: **DEMO Tree**

Conventional Preparation

```
BLK FORM
TOOL CALL 1 . . .
```

SL II Cycles

```
CYCL DEF 14.0 CONT. GEOM.
CYCL DEF 14.1 CONT. LABEL 1
CYCL DEF 20.0 CONTOUR DATA
CYCL DEF 22.0 ROUGH OUT
```

First datum

```
CYCL DEF 7.0 DATUM SHIFT
```

Call contour

```
CYCL CALL . . .
```

Second datum
Call contour
•
•
•

Retract tool, end *L Z100 . . . M2*

Contour, SPGM 1

```
LBL 1
•
•
•
•
•
•
•
LBL 0
```

Set up datum table

```
BEGIN 276    .D MM
D        X          Y          Z          C          A
0        +0         +0         +0         +0         +0
1        +20        +30        +0         +0         +0
2        •          •          •          •          •
3        •          •          •          •          •
4        •          •          •          •          •
•        •          •          •          •          •
•        •          •          •          •          •
•        •          •          •          •          •
•        •          •          •          •          •
END
```

Solution: **DEMO Tree**





P0 ≠ Contour point

Point	X	Y	Point	X	Y
P0	0,0	0,0	M1	-5,5	13,5
P1	–	9,0	M2	2,5	24,0
P2	–	10,0	M3	12,0	13,0
			M4	–	-2,0
			M5	-3,0	-18,0
			M6	8,0	-18,0

Main program

```
0 BEGIN PGM 276 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-10
2 BLK FORM 0.2 X+130 Y+170 Z+0
3 TOOL CALL 3 Z S3000 ..... R1,5
4 L Z+100 R0 F MAX
5 L X+0 Y+0 R0 F9999
6 L Z+2 M3
7 CYCL DEF 14.0 CONTOUR GEOMETRY
8 CYCL DEF 14.1 CONTOUR LABEL 1
9 CYCL DEF 20.0 CONTOUR DATA
  Q1=-5 ..... MILLING DEPTH
  Q2=1 ..... TOOL PATH OVERLAP
  Q3=+0 ..... ALLOWANCE FOR SIDE
  Q4=+0 ..... ALLOWANCE FOR FLOOR
  Q5=+0 ..... WORKPIECE SURFACE COORD.
```

```
    Q6=2 ..... SET UP CLEARANCE
    Q7=+5 ..... CLEARANCE HEIGHT
    Q8=0 ..... ROUNDING RADIUS
    Q9=-1 ..... DIRECTION OF ROTATION
10  CYCL DEF 22.0 ROUGH OUT
    Q10=5 ..... PLUNGING DEPTH
    Q11=100 ..... FEED RATE FOR PLUNGING
    Q12=200 ..... FEED RATE FOR MILLING
    Q18=0 ..... COARSE ROUGHING TOOL
    Q19=150 ..... RECIPROCATION FEED RATE
11  CYCL DEF 7.0 DATUM SHIFT
12  CYCL DEF 7.1 #1
13  CYCL CALL
14  CYCL DEF 7.0 DATUM SHIFT
15  CYCL DEF 7.1 #2
16  CYCL CALL
17  CYCL DEF 7.0 DATUM SHIFT
18  CYCL DEF 7.1 #3
19  CYCL CALL
20  CYCL DEF 7.0 DATUM SHIFT
21  CYCL DEF 7.1 #4
22  CYCL CALL
23  CYCL DEF 7.0 DATUM SHIFT
24  CYCL DEF 7.1 #5
25  CYCL CALL
26  CYCL DEF 7.0 DATUM SHIFT
27  CYCL DEF 7.1 #6
28  CYCL CALL
29  CYCL DEF 7.0 DATUM SHIFT
30  CYCL DEF 7.1 #7
31  CYCL CALL
32  CYCL DEF 7.0 DATUM SHIFT
33  CYCL DEF 7.1 #8
34  CYCL CALL
```

Retract tool, end

```
35  L Z+100 R0 F MAX M2
```

SPGM

```
36  LBL 1
37  L X+2,5 Y+32,5 RR
38  FC DR- R8,5 CCX+2,5 CCY+24
39  FC Y+10 DR- R7,5 CCX+12 CCY+13
40  FSELECT 2
41  FC DR- R40 CCY-2
42  FSELECT 2
43  FCT DR+ R10
44  FCT DR- R2 CCX+8 CCY-18
45  FSELECT 1
46  FCT DR+ R8
47  FCT DR- R2 CCX-3 CCY-18
48  FSELECT 2
49  FCT Y+9 DR+ R36
50  FC DR- R6,5 CCX-5,5 CCY+13,5
51  FC X+2,5 Y+32,5 DR- R8,5 CCX+2,5 CCY+24
52  FSELECT 2
53  LBL 0
54  END PGM 276 MM
```

Solution: **DEMO Tree**

Datum table

BEGIN 276 .D MM

D	X	Y	Z	C	A
0	+0	+0	+0	+0	+0
1	+20	+30	+0	+0	+0
2	+60	+30	+0	+0	+0
3	+100	+30	+0	+0	+0
4	+40	+80	+0	+0	+0
5	+80	+80	+0	+0	+0
6	+20	+130	+0	+0	+0
7	+60	+130	+0	+0	+0
8	+100	+130	+0	+0	+0
9	+0	+0	+0	+0	+0
10	+0	+0	+0	+0	+0
11	+0	+0	+0	+0	+0
12	+0	+0	+0	+0	+0
13	+0	+0	+0	+0	+0
14	+0	+0	+0	+0	+0
15	+0	+0	+0	+0	+0
16	+0	+0	+0	+0	+0
17	+0	+0	+0	+0	+0
18	+0	+0	+0	+0	+0
19	+0	+0	+0	+0	+0
20	+0	+0	+0	+0	+0

[END]

Task: **Datum shift and mirror images** Program(s): _____

_____



BEGIN PGM . . . MM
⋮

Datum shift
Part ① SPGM-Call

Part ② Mirror images
SPGM-Call

Part ③ Mirror images
SPGM-Call

Part ④ Mirror images
SPGM-Call

Reset cycle

Retract tool, end *L Z +100 R0 F9999 M2*

SPGM

LBL . . .

SPGM end

LBL 0

Solution: **Datum shift and mirror images**



MAIN PROGRAM

```
0  BEGIN PGM 229 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-20
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 7 Z S4000 ..... R4
4  L Z+100 R0 F MAX
5  CYCL DEF 7.0 DATUM SHIFT
6  CYCL DEF 7.1 X+50
7  CYCL DEF 7.2 Y+50
8  CALL LBL 1

9  CYCL DEF 8.0 MIRROR IMAGE
10 CYCL DEF 8.1 X
11 CALL LBL 1

12 CYCL DEF 8.0 MIRROR IMAGE
13 CYCL DEF 8.1 Y
14 CALL LBL 1

15 CYCL DEF 8.0 MIRROR IMAGE
16 CYCL DEF 8.1 X Y
17 CALL LBL 1
```

Retract tool, end

```
18 L Z+100 R0 F MAX M2
```

SPGM, Contour

```
19 LBL 1
20 L X+0 Y+0 R0 F MAX  M3 ..... AUXILIARY POINT R0
21 L Z+2 F MAX
22 L Z-15 R0 F100 ..... PLUNGING DEPTH
23 APPR LCT X+10 Y+10 R5 RL F200 ..... APPROACH STARTING POINT
                                       OF CONTOUR TANGENTIALLY
24 L Y+45
25 L X+30 Y+10
26 L X+10 Y+10
27 DEP LCT X+0 Y+0 R5 ..... DEPART TANGENTIALLY
28 LBL 0
29 END PGM 229 MM
```

Task: **Coordinate transformations**

Program(s): _____





MAIN PROGRAM

```
0 BEGIN PGM 232 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S4000 ..... R3
4 L Z+100 R0 F MAX M3
5 CYCL DEF 7.0 DATUM SHIFT
6 CYCL DEF 7.1 X+20
7 CYCL DEF 7.2 Y+10
8 CALL LBL 1

9 CYCL DEF 7.0 DATUM SHIFT
10 CYCL DEF 7.1 X+60
11 CYCL DEF 7.2 Y+10
12 CYCL DEF 11.0 SCALING
13 CYCL DEF 11.1 SCL1,2
14 CALL LBL 1

15 CYCL DEF 11.0 SCALING
16 CYCL DEF 11.1 SCL1
17 CYCL DEF 7.0 DATUM SHIFT
18 CYCL DEF 7.1 X+20
19 CYCL DEF 7.2 Y+60
20 CYCL DEF 10.0 ROTATION
21 CYCL DEF 10.1 ROT+20
22 CALL LBL 1

23 CYCL DEF 10.0 ROTATION
24 CYCL DEF 10.1 ROT+0
25 CYCL DEF 7.0 DATUM SHIFT
26 CYCL DEF 7.1 X+90
27 CYCL DEF 7.2 Y+60
28 CYCL DEF 8.0 MIRROR IMAGE
29 CYCL DEF 8.1 X
30 CALL LBL 1
```

```
31 CYCL DEF 8.0 ROTATION
32 CYCL DEF 8.1
33 CYCL DEF 7.0 DATUM SHIFT
34 CYCL DEF 7.1 X+0
35 CYCL DEF 7.2 Y+0
```

Retract tool, end

```
36 L Z+100 R0 F MAX M2
```

SPGM, Contour

```
37 LBL 1
38 L X+10 Y+10 R0 F MAX M3
39 L Z+2 F MAX
40 L Z-15 R0 F100
41 APPR LCT X+0 Y+10 R2 RR
42 L Y+20 X+0
43 RND R4
44 L X+20 Y+20
45 RND R4
46 L Y+0
47 CC X+15 Y+0
49 C X+10 Y+0 DR-
49 L X+0 Y+0
50 RND R4
51 L X+0 Y+10
52 DEP LCT X+10 Y+10 R2
53 L Z+2 R0 F MAX
54 LBL 0
55 END PGM 232 MM
```

Task: **Scaling factor – hammer**

Program(s): _____



Solution: **Scaling factor – hammer**



Main program

0 BEGIN PGM 284 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-10
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 6 Z S2000 ..... R3
4 CYCL DEF 7.0 DATUM SHIFT
5 CYCL DEF 7.1 X+5
6 CYCL DEF 7.2 Y+5
7 CALL LBL 1

8 CYCL DEF 7.0 DATUM SHIFT
9 CYCL DEF 7.1 X+95
10 CYCL DEF 7.2 Y+15
11 CYCL DEF 10.0 ROTATION
12 CYCL DEF 10.1 ROT+105
13 CALL LBL 1

14 CYCL DEF 10.0 ROTATION
15 CYCL DEF 10.1 ROT+0
16 CYCL DEF 7.0 DATUM SHIFT
17 CYCL DEF 7.1 X+20
18 CYCL DEF 7.2 Y+95
19 CYCL DEF 8.0 MIRROR IMAGE
20 CYCL DEF 8.1 Y
21 CYCL DEF 26.0 AXIS-SPEC. SCALING
22 CYCL DEF 26.1 X1,9 Y1 CCX+10 CCY+95
23 CALL LBL 1

24 CYCL DEF 26.0 AXIS-SPEC. SCALING
25 CYCL DEF 26.1 X1 Y1 CCX+0 CCY+0
26 CYCL DEF 8.0 MIRROR IMAGE
27 CYCL DEF 8.1
28 CYCL DEF 7.0 DATUM SHIFT
29 CYCL DEF 7.1 X+0
30 CYCL DEF 7.2 Y+0

31 L Z+100 R0 F MAX M2

Solution: **Scaling factor – hammer**

SPGM

```
32  LBL 1
33  L X+7 Y+15 R0 F300
34  L Z-5
35  APPR LCT X+0 Y+15 R1 RR
36  L Y+25
37  L X+5 Y+40
38  L X+10 Y+25
39  L Y+20
40  L X+40
41  L Y+10
42  L X+10
43  L Y+0
44  L X+0
45  L Y+15
46  DEP LCT X+8 Y+15 R1
47  L Z+2 R0
48  LBL 0
49  END PGM 284 MM
```

Task: **Scaling factor**

Program(s): _____



Solution: **Scaling factor**



MAIN PROGRAM

```
0 BEGIN PGM 234 MM
1 BLK FORM 0.1 Z X+0 Y+0 Z-20
2 BLK FORM 0.2 X+100 Y+100 Z+0
3 TOOL CALL 8 Z S4000 ..... R5
4 L Z+100 R0 F MAX M3
5 CALL LBL 1 ..... SPGM-CALL PART ①

6 CYCL DEF 7.0 DATUM SHIFT
7 CYCL DEF 7.1 X+50
8 CYCL DEF 7.2 Y+50
9 CYCL DEF 26.0 AXIS-SPEC. SCALING
10 CYCL DEF 26.1 X1,5 CCX+10
11 CALL LBL 1 ..... SPGM-CALL PART ②

12 CYCL DEF 7.0 DATUM SHIFT
13 CYCL DEF 7.1 X+0
14 CYCL DEF 7.2 Y+0
15 CYCL DEF 26.0 AXIS-SPEC. SCALING
16 CYCL DEF 26.1 X1 ..... CYCLE RESET
```

Retract tool, end

```
17 L Z+100 R0 F MAX M2
```

SPGM, Contour

```
18 LBL 1
19 L X+0 Y+0 R0 F MAX
20 L Z+2 F MAX
21 L Z-10 R0 F100 M8
22 APPR LT X+10 Y+10 LEN5 RL F200
23 L IY+20
24 L IX+21
25 L IX+4 IY-4
26 L Y+10
27 L X+10
28 DEP LT LEN5
29 L X+0 Y+0 R0
30 L Z+2 R0 F MAX
31 LBL 0
32 END PGM 234 MM
```

Task: **Tilt working plane**

Program(s): _____





Solution: **Tilt working plane**



Complete program

```
0  BEGIN PGM 285 MM
1  BLK FORM 0.1 Z X+0 Y+0 Z-40
2  BLK FORM 0.2 X+100 Y+100 Z+0
3  TOOL CALL 19 Z S4000 ..... CALL TOOL DATA, L ABSOLUTE
                              IN TOOL TABLE!
4  CYCL DEF 1.0 PECKING ..... DEFINE CYCLE
5  CYCL DEF 1.1 SET UP +2
6  CYCL DEF 1.2 DEPTH -20
7  CYCL DEF 1.3 PECKG-5
8  CYCL DEF 1.4 DWELL 0
9  CYCL DEF 1.5 F400

10 L Z+100 R0 F MAX ..... RETRACT TOOL AXIS
11 L X+100 Y+50 R0 F MAX ..... PRE-POSITIONING
12 L B+27 R0 F MAX ..... TILT HEAD (MOVE)

13 CYCL DEF 19.0 WORKING PLANE
14 CYCL DEF 19.1 B+27 ..... TILT HEAD (CALCULATE)
15 L X+50 Y+50 R0 F MAX M3 ..... MOVE TO DRILLING POSITION
16 L Z+2 R0 F MAX M99 ..... Z-CLEARANCE, CALL CYCLE 1
17 CYCL DEF 19.0 WORKING PLANE
18 CYCL DEF 19.1 B+0 ..... TILT RESET

19 L Z+100 R0 F MAX M5 ..... RETRACT TOOL
20 L B+0 R0 F MAX M30 ..... TILT HEAD BACK
21 END PGM 285 MM
```

Task: **Tilt working plane**

Program(s): _____





Solution: **Tilt working plane**



Complete program

```
0 BEGIN PGM 286 MM
1 BLK FORM 0.1 Z X-50 Y-50 Z-40
2 BLK FORM 0.2 X+50 Y+50 Z+0
3 TOOL CALL 19 Z S4000 ..... CALL TOOL DATA, L ABSOLUTE
                             IN TOOL TABLE!
4 CYCL DEF 1.0 PECKING ..... DEFINE CYCLE
5 CYCL DEF 1.1 SET UP +2
6 CYCL DEF 1.2 DEPTH -20
7 CYCL DEF 1.3 PECKG -5
8 CYCL DEF 1.4 DWELL 0
9 CYCL DEF 1.5 F400
10 L Z+100 R0 F MAX ..... RETRACT TOOL AXIS
11 L X+150 R0 F MAX ..... PRE-POSITIONING
12 CYCL DEF 7.0 DATUM SHIFT
13 CYCL DEF 7.1 X+50 ..... ZERO POINT ON RIGHT EDGE
14 L B+90 R0 F MAX ..... TILT HEAD (MOVE)

15 CYCL DEF 19.0 WORKING PLANE
16 CYCL DEF 19.1 B+90 ..... TILT HEAD (CALCULATE)
17 L X+25 Y+0 R0 F MAX M3 ..... MOVE TO DRILLING POSITION
18 L Z+2 R0 F MAX M99 ..... Z-CLEARANCE, CALL CYCLE 1
19 CYCL DEF 19.0 WORKING PLANE
20 CYCL DEF 19.1 B+0 ..... TILT RESET

21 CYCL DEF 7.0 DATUM SHIFT
22 CYCL DEF 7.1 X+0
23 L Z+100 R0 F MAX M5 ..... RETRACT TOOL
24 L B+0 R0 F MAX M30 ..... TILT HEAD BACK
25 END PGM 286 MM
```